

The Padovan sequence is similar to the Fibonacci sequence in several ways.
Some are given in the table below, and the referenced video shows some of the geometric
similarities.



 Comment 
 Padovan 
 Fibonacci


 
 



 Named after. 
 Richard Padovan 
 Leonardo of Pisa: Fibonacci


 
 



 Recurrence initial values. 
 P(0)=P(1)=P(2)=1 
 F(0)=0, F(1)=1


 Recurrence relation. 
 P(n)=P(n-2)+P(n-3) 
 F(n)=F(n-1)+F(n-2)


 
 



 First 10 terms. 
 1,1,1,2,2,3,4,5,7,9 
 0,1,1,2,3,5,8,13,21,34


 
 



 Ratio of successive terms... 
 Plastic ratio, p 
 Golden ratio, g


 
 1.324717957244746025960908854… 
 1.6180339887498948482...


 Exact formula of ratios p and q. 
 ((9+69**.5)/18)**(1/3) + ((9-69**.5)/18)**(1/3) 
 (1+5**0.5)/2


 Ratio is real root of polynomial. 
 p: x**3-x-1 
 g: x**2-x-1


 
 



 Spirally tiling the plane using. 
 Equilateral triangles 
 Squares


 
 



 Constants for ... 
 s= 1.0453567932525329623 
 a=5**0.5


 ... Computing by truncation. 
 P(n)=floor(p**(n-1) / s + .5) 
 F(n)=floor(g**n / a + .5)


 
 



 L-System Variables. 
 A,B,C 
 A,B


 L-System Start/Axiom. 
 A 
 A


 L-System Rules. 
 A->B,B->C,C->AB 
 A->B,B->AB

Task

 Write a function/method/subroutine to compute successive members of the Padovan series using the recurrence relation.
 Write a function/method/subroutine to compute successive members of the Padovan series using the floor function.
 Show the first twenty terms of the sequence.
 Confirm that the recurrence and floor based functions give the same results for 64 terms,
 Write a function/method/... using the L-system to generate successive strings.
 Show the first 10 strings produced from the L-system
 Confirm that the length of the first 32 strings produced is the Padovan sequence.
Show output here, on this page.

Ref

 The Plastic Ratio - Numberphile video.


with javascript_semantics
sequence padovan = {1,1,1}
function padovanr(integer n)
    while length(padovan)<n do
        padovan &= padovan[$-2]+padovan[$-1]
    end while
    return padovan[n]
end function
 
constant p = 1.324717957244746025960908854,
         s = 1.0453567932525329623 
function padovana(integer n)
    return floor(power(p,n-2)/s + 0.5)
end function
 
constant l = {"B","C","AB"}
function padovanl(string prev)
    string res = ""
    for i=1 to length(prev) do
        res &= l[prev[i]-64]
    end for
    return res
end function
 
sequence pl = "A", l10 = {}
for n=1 to 64 do
    integer pn = padovanr(n)
    if padovana(n)!=pn or length(pl)!=pn then crash("oops") end if
    if n<=10 then l10 = append(l10,pl) end if
    pl = padovanl(pl) 
end for
printf(1,"The first 20 terms of the Padovan sequence: %v\n\n",{padovan[1..20]})
printf(1,"The first 10 L-system strings: %v\n\n",{l10})
printf(1,"recursive, algorithmic, and l-system agree to n=64\n")
